

Przedmiot: Informatyka w inżynierii produkcji

Forma: Laboratorium

Temat: Zadanie 4. Instrukcja warunkowa.

Celem ćwiczenia jest nabycie umiejętności wykorzystania w praktyce instrukcji warunkowych programowania VBA. Zadanie do wykonania polega na utworzeniu uproszczonej automatycznej ewidencji towarów na magazynie z wykorzystaniem podprogramów napisanych w języku VBA.


Instrukcja warunkowa IF

Postać wierszowa

```
If warunek Then polecenie [:polecenie_1]...[:polecenie_n]
```

Postać blokowa

```
If warunek Then  
    [polecenie_1]  
    ...  
    [polecenie_n]  
End If
```




Instrukcja warunkowa IF ...ELSE

Postać wierszowa

```
If warunek Then [blok_poleceń_1] Else [blok_poleceń_2]
```

Postać blokowa

```
If warunek Then  
    [polecenie_1]  
Else  
    [polecenie_n]  
End If
```



Instrukcje IF można zagnieżdżać.

Instrukcja WITH ...END WITH

Wykonuje szereg instrukcji, które wielokrotnie odwołują się do pojedynczego obiektu lub struktury, dzięki czemu instrukcje mogą używać uproszczonej składni.

```
With obiekt  
    [ .właściwość ]  
End With
```

Przykład:

```
With ActiveCell.Font  
    .Bold = True  
    .Color = vbBlue  
    .Name = "Arial"  
    .Size = 22  
    .Italic = True  
End With
```

CHARAKTERYSTYKA ZADANIA

Zadanie polega na wykorzystaniu arkuszy kalkulacyjnych Excel do utworzenia narzędzia uproszczonej ewidencji towarów w magazynie. Narzędzie będzie się składało z dwóch arkuszy kalkulacyjnych:

- arkusz „Zestawienie towarów”, które ma na celu przechowywanie aktualnej informacji o towarach i ich stanach magazynowych oraz ostatniej transakcji (kupna/sprzedaży) związanej z danymi towarami,
- arkusz „Historia transakcji” służy do przechowywania historii dotyczącej wykonywanych transakcji na danym towarze.

ZADANIA DO WYKONANIA

1. Otwórz pusty plik Excel i zapisz go z rozszerzeniem umożliwiającym pracę z Makrami (*.xlsm).
2. Przygotuj arkusz kalkulacyjny „Zestawienie towarów” według poniższego wzoru.

The screenshot shows an Excel spreadsheet with the following data:


Lp.	Nazwa towaru	Jednostka miary	Data ostatniej zmiany	Sprzedaż	Zakup	Stan magazynu
1	Równoważnia R50	szt				12
2	Równoważnia R60	szt				18
3	Równoważnia R70	szt				5
4	Drabinka rehabilitacyjna 220x90	szt				12
5	Drabinka rehabilitacyjna 245x90	szt				10
6	Drabinka rehabilitacyjna 220x75	szt				9
7	Drabinka rehabilitacyjna 245x75	szt				20
8	Lustro korekcyjne	szt				2
9	Ławeczka 100 cm	szt				5
10	Ławeczka 200 cm	szt				5

3. Przygotuj arkusz kalkulacyjny „Historia transakcji” wg poniższego wzoru.

The screenshot shows an Excel spreadsheet with the following data:

Lp.	Nazwa towaru	Jednostka miary	Data	Liczba	Rodzaj transakcji

4. Utwórz makro, które zautomatyzuje ewidencje towarów poprzez wpisywanie właściwego rekordu do historii. W tym celu kliknij prawym przyciskiem myszy etykietę arkusza „Zestawienie towarów” i z menu podręcznego wybierz „Wyświetl kod”. W edytorze VBA wybierz z rozwijanej listy pozycję *Worksheet* (z lewej strony) i *Change* (z prawej strony). Automatycznie zostanie wybrany właściwy kod jak poniżej.



Wpisany kod oznacza procedurę, która jest uruchamiana wówczas, gdy ktoś dokona zmiany w jakiegokolwiek komórce arkusza „Zestawienie towarów”.

5. Uzupełnij procedurę *Worksheet_Change* właściwym kodem VBA (jak poniżej).

```
Private Sub Worksheet_Change(ByVal Target As Range)
    'Target ->obszar arkusza w którym nastąpiła zmiana

    '----deklaracja zmiennych
    ' zmienna służąca do sprawdzenia poprawności warunków działania makra
    Dim poprawnosc As Integer
    'zmienna służąca do zmiany stanu towaru
    Dim stan As Double
    'zmienna służąca do wyszukania pierwszego wolnego wiersza w arkuszu z Historią
    Dim ostatni As Long
    '----koniec deklaracji zmiennych

    '----właściwy kod makra
    With Target

        poprawnosc = IsNumeric(.Value) * (.Cells.Count = 1) * _
            (.Column < 7) * (.Column > 4) * (.Row > 3)

        If poprawnosc Then 'sprawdzenie warunku poprawności

            'ustalenie aktualnego stanu towaru
            stan = Cells(.Row, 7).Value + .Value * (-1) ^ (.Column)

            'dodanie zmiany stanu do arkusza z historią, jeżeli stan>=0
            If stan >= 0 Then

                'zmiana stanu w kolumnie 7 i wierszu w którym nastąpiła zmiana
                Cells(.Row, 7).Value = stan
                'zmiana daty na aktualną w kolumnie 4 i wierszu w którym nastąpiła zmiana
                Cells(.Row, 4).Value = Date

                'wpis do arkusza z historią wiersza dotyczącego zmiany
                With Worksheets("Historia transakcji")
                    'zmiennej ostatni nadajemy wartość równą numerowi wiersza,
                    'w którym będziemy wpisywać dane
                    ostatni = .Cells(65536, 2).End(xlUp).Row + 1


                    .Cells(ostatni, 2) = Cells(Target.Row, 2).Value
                    .Cells(ostatni, 3) = Cells(Target.Row, 3).Value
                    .Cells(ostatni, 4) = Cells(Target.Row, 4).Value
                    .Cells(ostatni, 5) = Target.Value
                    If Target.Column = 5 Then
                        .Cells(ostatni, 6) = "sprzedaż"
                    Else
                        .Cells(ostatni, 6) = "kupno"
                    End If
                End With

            End With
        End If

    End With
    '----koniec właściwego kodu makra

End Sub
```

Szerszego komentarza wymaga sposób ustalania zmiennej *poprawnosc*. Stanowi ona iloczyn wartości typu Prawda lub Fałsz. Gdy choć jedno wyrażenie jest nie prawdziwe to zmienna przybiera wartość Fałsz, co powoduje, iż nie wykonana się dalsza część procedury.



Można zauważyć, iż każdy algorytm można zapisać w postaci odmiennego kodu programu na przykład zamiast wierszy:

```

If Target.Column = 5 Then
    .Cells(ostatni, 6) = "sprzedaż"
Else
    .Cells(ostatni, 6) = "kupno"
End If
  
```

wystarczy napisać jedynie linijkę:

```

.Cells(ostatni, 6) = IIf(Target.Column = 5, "sprzedaż", "kupno")
  
```

- Przetestuj działanie makra - wpisz parę zmian w stanie magazynowym (jedynie w kolumnach E i F). Kolumna z datą (D) wypełniana jest automatycznie. Przykład poprawnego działania zamieszczono poniżej.

	A	B	C	D	E	F	G
1	Zestawienie towarów na magazynie						
2							
3	Lp	Nazwa towaru	Jednostka miar	Data ostatniej zmian	Sprzedaż	Zakup	Stan magazynu
4	1	Równoważnia R50	szt	2014-09-17	6		6
5	2	Równoważnia R60	szt				18
6	3	Równoważnia R70	szt	2014-09-17		2	7
7	4	Drabinka rehabilitacyjna 220x90	szt				12
8	5	Drabinka rehabilitacyjna 245x90	szt				10
9	6	Drabinka rehabilitacyjna 220x75	szt	2014-09-17	2		7
10	7	Drabinka rehabilitacyjna 245x75	szt	2014-09-17	2		18
11	8	Lustro korekcyjne	szt				2
12	9	Ławeczka 100 cm	szt				5
13	10	Ławeczka 200 cm	szt	2014-09-17		1	6
14							

	A	B	C	D	E	F
1	Historia transakcji towarów na magazynie					
2						
3	Lp.	Nazwa towaru	Jednostka miary	Data	Liczba	Rodzaj transakcji
4		Równoważnia R50	szt	2014-09-17		6 kupno
5		Równoważnia R70	szt	2014-09-17		2 sprzedaż
6		Drabinka rehabilitacyjna 220x75	szt	2014-09-17		2 kupno
7		Drabinka rehabilitacyjna 245x75	szt	2014-09-17		2 kupno
8		Ławeczka 200 cm	szt	2014-09-17		1 sprzedaż
9						
10						

7. *Działanie makra nie jest „idealne”. Dopisz odpowiedni kod VBA programu, aby w rekordzie historii transakcji (arkusz „Historia transakcji”) pojawiła się odpowiednia liczba porządkowa (kolumna A).*
8. *Można zauważyć, iż w zestawieniu towarów, jeżeli ostatnia zmiana dotyczyła sprzedaży, w kolumnie E znajduje się odpowiednia liczba sprzedanych towarów. Jeżeli teraz zakupimy ten sam towar, wpiszemy odpowiednią liczbę w kolumnie F – kolumnie zakupu. Niestety liczba poprzedniej sprzedaży nie znika, zatem teraz nie wiadomo czy ostatnia transakcja dotyczyła sprzedaży czy zakupu. Dodaj odpowiedni kod VBA do makra, tak aby przy wpisywaniu liczby sprzedaży, zniknęła liczba zakupu dla tego samego towaru i na odwrót (przy wpisywaniu liczby zakupu, zniknęła liczba sprzedaży).*
9. *Popraw działanie makra tak, aby nie można było wpisać (zapisać) wartości sprzedaży towaru w liczbie większej niż stan na magazynie. Powinno się także pojawić odpowiednie okienko z komunikatem informacyjnym.*

Literatura

Gromulski Piotr (Red.), 35 najlepszych narzędzi i makr, Wydawnictwo Wiedza i Praktyka, Warszawa 2011.